SECURITY PROJECTS

for the TRS-80 COLOR COMPUTER

INTRODUCTORY

Projects given in this Booklet were designed around the TRS 80 Color Computer Two. The Software was written using the Extended Basic. LET statements were inserted to enhance the understanding of the Programs. Parts for these Security Projects were bought from Radio Shack, Mouser Electronics, and Jameco Electronics. Addresses for the last two Vendors are given below.

MOUSER ELECTRONICS
2401 Hwy 287 North
Mansfield, Texas 76063
Phone: (817) 483-4422

JAMECO ELECTRONICS
1355 Shoreway Road
Belmont, CA 94002
Phone: (415) 592-8097

TABLE OF CONTENT

A PRODUCT OF

BROWN'S ENTERPRISES
119 Skyline Dr. R.H.
Granbury, Texas 76048
Published 1987

BROWN'S ENTERPRISES disclaims any responsibility for the safe and proper functioning of Reader-Built Projects based upon or from plans or Information published in this Booklet.

VACATION LIGHT

The Vacation Light Circuit consist of two parts, which connect into the left Joystick Jack and Cassette Jack, as shown in Figure 3. J1,R1,R2, and P1 make up the Light Sensor. The Potentiometer R2, is the sensitivity control. R2 is adjusted to the light level which corresponds to 33, for the Joystick input or approximately 2.5 volts, across the Photocell P1. The Color Computer Two interpretates the 2.5 VDC as a 33. The Joystick analog inputs only have 64 discrete levels. The Lamp Control Circuit consist of J3,R3,R4,TR1,IC1,IC2 and T1. When the motor on command is read, a switch contact is closed inside the computer, turning on the L.E.D. inside the Optical Isolator IC1. The L.E.D. activates a optical coupled Triac, which fires the 400 VAC Triac TR1, in turn energizing the Lamp (L1 or RL).

The Program is fairly straight forward. You type in the number of hours, minutes and seconds which the Lamp is to stay on. When the light level drops to where the Joystick input is, 33 or approximately 2.5 Volts across P1, the Lamp is turned on. A Timer Routine will convert the on time you entered to seconds, and will count down to Zero, at which time the Lamp is cut off. The Program will wait until daylight before resetting.

The Circuits can be constructed with any wiring technique. You can etch your own boards or use predrilled Perf Board. When soldering the cable to the Din-Connectors, be sure to follow the Jack Pin connections, as shown in Figure 3. During the initial calibration, adjust R2 until it is no longer in the circuit. Use a Book as a shade to test the function of the Program and Circuit. When an object is over the Photocell, the Computer thinks it is Dark. After finishing with the testing, adjust R2 to the desired sensitivity. Place the Photocell in a location which will not allow it to be falsely triggered or reset from car head lamps, animals, etc.

VACATION LIGHT PROGRAM

```
10  PRINT "HOW LONG DO YOU WANT THE LIGHT(S) ON?"
20  PRINT "HOURS, MINUTES, SECONDS";
30  INPUT H,M,S
40  CLS
45  REM  MONITORING LIGHT LEVEL (TRIP).
50  LET L = JOYSTK (0)
60  IF L < 33 THEN 50
65  REM  LAMP IS TURNED ON.
70  MOTOR ON
80  PRINT "TIMER ON."
85  REM  THE HOURS, MINUTES, SECONDS, ENTERED ARE CONVERTED
    TO TOTAL SECONDS.
90  LET S = H * 3600 + M * 60 + S
95  REM  COUNT DOWN COUNTER.
100  FOR Z = S TO 0 STEP - 1
110  FOR T = 1 TO 375
120  NEXT T
130  NEXT Z
135  REM  LAMP IS TURNED OFF.
140  MOTOR OFF
150  PRINT "TIMER OFF."
155  REM  MONITORING LIGHT LEVEL (RESET).
160  LET L = JOYSTK (0)
170  IF L < 25 THEN 40 ELSE 160
```



VACATION LIGHT CIRCUIT

FIG. 3

VACATION LIGHT

PARTS LIST

	RESISTORS	RADIO SHACK NO.
R1	10K OHM ½ WATT	271 - 034
R3	1K OHM ½ WATT	271 - 023
R4	180 OHM ½ WATT	271 - 014
R2	100K OHM POTENTIOMETER	271 - 338
P1	Cds PHOTOCELL (PHOTORESISTOR)	276 - 116
	ICS	
IC1	MOC3010 OPTICAL COUPLER TRIAC OUTPUT	276 - 134
IC2	1-AMP 50 PIV DIODE BRIDGE	276 - 1161
	SEMICONDUCTORS	
TR1	400 VAC 6 AMP TRIAC	276 - 1000
	MISCELLANEOUS	
J1	5 PIN DIN-TYPE CONNECTOR	274 - 003
J3	6 PIN DIN-TYPE CONNECTOR	274 - 020
T1	12.6 VAC 300MA MINIATURE TRANSFORMER	273 - 1385
L1	40 WATT HOUSEHOLD LAMP	
	CABLE 4 CONDUCTOR CABLE	

TEMPERATURE ALARM

The Temperature Alarm is a fairly simple device. The system is comprised of two independent circuits, as shown in Figure 4. The Thermistor Circuit is connected into the left Joystick Jack via a Din-Connector. The Two-Tone Piezo Buzzer Circuit connects into the Cassette Jack via a Din-Connector also. The Program is very simple and only uses a few lines. It allows the Computer to monitor the Joystick (0) Input and can be modified to monitor four separate Thermistors, see Program. When the Computer's Joystick input reaches 32, the Motor On Command is performed. 32 is the Color Computer Two's interpitation of the voltage drop across the Thermistor. Potentiometer R1 is adjusted for the resistance which corresponds to the selected temperature alarm point. The R1 Potentiometer resistance can be derived from the formulas and examples in Table 1.

You can test the function of the Circuit and Program by placing the Thermistor in warm water. Make sure that R1 is set to the desired resistance, corresponding to a temperature less than the water. The Buzzer should sound if everything is working. The Alarm can be cleared by pushing the "C" Key. This Program can also be used as a Low Temperature Alarm, see Program. The circuit can be constructed with any wiring technique. You can etch your own Board or use Predrilled Perf Board. When soldering the Cable to the Din-Connectors, be sure to follow the Jack Pin Connections, as shown in Figure 4.

TEMPERATURE ALARM

PROGRAM

```
10  CLS
15  REM  MONITOR TEMPERATURE SENSOR.
20  LET P = JOYSTK (O)
30  IF  P < = 32  THEN MOTOR ON ELSE 20
35  REM  LOOKING FOR RESET.
40  LET A$ = INKEY$
50  IF A$ = " C " THEN MOTOR OFF ELSE 40
60  GOTO  20
```

INSERT THE FOLLOWING INTO LINE 30 TO CHANGE THE PROGRAM TO A LOW TEMPERATURE ALARM. NOTICE, YOU ARE ONLY CHANGING THE LESS THAN SYMBOL.

```
IF  P > = 32  THEN MOTOR ON ELSE 20
```

INSERT THE FOLLOWING PROGRAM LINES TO MONITOR OTHER THERMISTORS THROUGH THE JOYSTK INPUTS.

```
20  LET  P1 = JOYSTK (O)
22  LET  P2 = JOYSTK (1)
24  LET  P3 = JOYSTK (2)
26  LET  P4 = JOYSTK (3)
30  IF P1 < = 32 OR P2 < = 32 OR P3 < = 32
    OR P4 < 32 THEN MOTOR ON ELSE 20
```

TEMPERATURE ALARM

TABLE

The following equation is used to convert the Temperature in Fahrenheit to Centigrade. The Centigrade Temperature is used in the Resistance Calculation Below.

Step 1 $°C = \frac{F - 32}{1.8}$

Step 2 $= \frac{50 - 32}{1.8}$

Step 3 $= \frac{18}{1.8}$

Step 4 °C = 10 or 10°C

If °F = 50
°C = ?
°F To °C

The following equation is used to calculate the approximate resistance of the Thermistor, at the desired Temperature. The exact Resistance can be found experimentally, by measuring the resistance of the Thermistor at the desired Temperature.

Step 1 $THR1 = Ae\ (\frac{B}{T})$

Step 2 $= 62.5\ (\frac{4100}{10})$

Step 3 $= 62.5\ (410)$

Step 4 THR1 = 25,625 ohm

B = 4100 ± 300
T = Temperature in Centigrade
Ae = 62.5
R = THR1 = ?



TEMPERATURE ALARM

FIG. 4

TEMPERATURE ALARM

PARTS LIST

RESISTORS

R1 500K OHM POTENTIOMETER	JAMECO ELECTRONICS NO. 43P-500K
R3,R4 4.7K OHM ½ WATT	RADIO SHACK NO. 271-030
THR1 NTC THERMISTOR	MOUSER ELECTRONICS NO. ME334-4127-103

ICs

IC1 1-AMP 50 PIV DIODE BRIDGE	RADIO SHACK NO. 276-1161

SEMICONDUCTORS

Q1 NPN TRANSISTOR 2N3053	RADIO SHACK NO. 276-2030

CAPACITORS

C1 1000 uF 35V RADIAL OR AXIAL	RADIO SHACK NO. 272-1032

MISCELLANEOUS

J1 5 PIN DIN-TYPE CONNECTOR	RADIO SHACK NO. 274-003
J3 6 PIN DIN-TYPE CONNECTOR	RADIO SHACK NO. 274-020
T1 12.6 VAC 300 mA MINIATURE TRANSFORMER	RADIO SHACK NO. 273-1385
B1 TWO-TONE PIEZO BUZZER	RADIO SHACK NO. 273-070

SOUND ACTIVATED ALARM

The Sound Sensor is a two stage Audio Amplifier, as shown in Figure 5, with the BiFET OP-AMPS directly coupled and Biased for a single supply. An Audio signal is rectified and filtered through D1,R7,C3, and feeds into the Joystick Jack Input. The Program is similar to the Temperature Alarm, except in line 30, where the 42 is used to set the sensitivity or trip point. See Program for further details. This Project also uses the same Piezo Buzzer Circuit as the Temperature Alarm, for the Audible Annunciator. The sound Alarm can be used to detect the Act of Breaking Glass, Baby Sitting, etc. There are no adjustments to be made in the Circuit itself, only in the Program, so once the construction is completed, the Device is ready to use.

The Circuit can be constructed with any wiring Technique. You can etch your own Board or use Predrilled Perf-Board. When Soldering the Cable to the Din-Connector, be sure to follow the Jack Pin Connections, as shown in Figure 5.

SOUND ACTIVATED

ALARM PROGRAM

```
10  CLS
15  REM  MONITORING AUDIO SENSOR.
20  P = JOYSTK (O)
30  IF P < 42 THEN 20 ELSE MOTOR ON
35  REM  WAITING FOR RESET.
40  LET  A$ = INKEY$
50  IF A$ = " C " THEN MOTOR OFF ELSE 40
60  GOTO 20
```

NOTE: THE NUMBER 42 IN LINE 30 IS THE SENSITIVITY CONTROL. TO INCREASE THE SENSITIVITY, CHANGE 42 TO A SMALLER VALUE SUCH AS 35. TO DECREASE THE SENSITIVITY, CHANGE THE 42 TO A LARGER VALUE SUCH AS 43. THE NUMBER YOU USE IS ARBITRARY.



SOUND ACTIVATED ALARM

FIG. 5

SOUND ACTIVATED ALARM

PARTS LIST

	RADIO SHACK NO.
RESISTORS	
R1,R5 1K OHM ½ WATT	271 - 023
R2 15K OHM ½ WATT	271 - 036
R3 10K OHM ½ WATT	271 - 034
R4,R6 100K OHM ½ WATT	271 - 045
R7 2.2 MEG OHM ½ WATT	271 - 061
CAPACITORS	
C1,C2,C3 2.2 uF	272 - 997
SEMICONDUCTOR	
D1 1N914 DIODE	276 - 1620
ICs	
IC1 DUAL-BIFET OP-AMP TL082	276 - 1715
MISCELLANEOUS	
M1 ELECTRET MIKE ELEMENT PC-MOUNT	270 - 090
J3 6-PIN DIN-CONNECTOR	274 - 020
CABLE 4 CONDUCTOR	

INTRUSION ALARM

The Intrusion Alarm is a Door/Window Security System, which has the capability of monitoring 16 Rooms. Each Room can be independently or simultaneously displayed when an intrusion is detected. The Circuit is basically a hardwired Digital to Analog Converter, as shown in Figure 2. Each Alarm Coder has four inputs or channels, and each channel has as many window/door switches as needed. The window/door switches are connected in parallel and are open contacts with magnets engaged. There is one Power Supply for all four Alarm Coders, as shown in Figure 1. Each Alarm Coder is connected to the 5 Volt Power Supply and there are two Coders to each Joystick input. This Project also uses the Alarm Buzzer, which is shown in the Temperature Alarm Schematic.

The Programs for the Intrusion Alarm are fairly straight forward. The example House Program was written to look at only one of the Alarm Coders for the first four channels. This Program shows how the color graphics could be used to display a house with each Room. When a particular Room comes into Alarm, it could be colored red or some other color. The Basic Intrusion Alarm Program does not use the Graphic instruction, but monitors all four Joystick Inputs. The two Programs could be combined and used to display a fairly sizeable house. The question marks, in the Print Statement are only used to take up space. They are to be replaced with the name of the Room being monitored, such as the kitch, Bedroom, etc. The time delay loop in Lines 20 and 30 allows the Computer time to detect several Intrusions, within 10 seconds. This time delay ensures detection of one or more simultaneous Intrusions. If the two Programs are combined, the Graphics section will have to be changed to display your particular house.

INTRUSION ALARM

CONTINUED

The circuit can be constructed with any wiring technique. You can etch your own Board or use Predrilled Perf-Board. When soldering the Cables to the Din-Connectors, be sure to follow the Jack Pin Connections, as shown in Figure 1 and 2. Be careful when wiring in the IC1 Chips, because the Pin Alignement of the 4069 is different from the 4049. There is a Pin Out Drawing enclosed which shows the ICs and Transistors.

The Intrusion Alarm could have many applications. The Security System could be used on a Fence or to monitor the Perimeter of some building. I am sure that the Builders will have no problem in coming up with their own Ideas.

BASIC INTRUSION ALARM

PROGRAM

```
10  CLS
15  REM  ALLOWS FOR DETECTION OF MORE THAN ONE
    ROOM ALARM IN 10 SECONDS.
20  FOR I = 1 TO 3750
30  NEXT I
35  REM  MONITORING OF FOUR DECODERS.
40  LET P1 = JOYSTK(0): LET P2 = JOYSTK(1):
    LET P3 = JOYSTK(2): LET P4 = JOYSTK(3)
50  LET T = P1 + P2 + P3 + P4
55  REM  ALARM ANNUNCIATES.
60  IF T > 0 THEN MOTOR ON ELSE 20
65  REM  ROOM P1 DECODER.
70  IF 33 < = P1 THEN PRINT "?": LET P1 = P1 - 33
80  IF 16 < = P1 THEN PRINT "?": LET P1 = P1 - 16
90  IF 7 < = P1 THEN PRINT "?": LET P1 = P1 - 7
100  IF 3 < = P1 THEN PRINT "?"
105  REM  ROOM P2 DECODER.
110  IF 33 < = P2 THEN PRINT "?": LET P2 = P2 - 33
120  IF 16 < = P2 THEN PRINT "?": LET P2 = P2 - 16
130  IF 7 < = P2 THEN PRINT "?": LET P2 = P2 - 7
140  IF 3 < = P2 THEN PRINT "?"
150  REM  ROOM P3 DECODER.
160  IF 33 < = P3 THEN PRINT "?": LET P3 = P3 - 33
170  IF 16 < = P3 THEN PRINT "?": LET P3 = P3 - 16
180  IF 7 < = P3 THEN PRINT "?": LET P3 = P3 - 7
190  IF 3 < = P3 THEN PRINT "?"
200  REM  ROOM P4 DECODER.
210  IF 33 < = P4 THEN PRINT "?": LET P4 = P4 - 33
220  IF 16 < = P4 THEN PRINT "?": LET P4 = P4 - 16
230  If 7 < = P4 THEn PRINT "?": LET P4 = P4 - 7
240  IF 3 < = P4 THEN PRINT "?"
245  REM  WAITING TO SILENCE ALARM
250  LET A$ = INKEY$
260  IF A$ = "S" THEN MOTOR OFF ELSE 250
265  REM  AFTER ALL ROOMS HAVE BEEN CHECKED, TYPE "C" TO RESET.
270  LET A$ = INKEY$
280  IF A$ = "C" THEN 10 ELSE 270
```

INTRUSION ALARM

EXAMPLE HOUSE PROGRAM

```
10  PMODE 3,1
20  PCLS
30  SCREEN 1,0
35  REM  DRAWING OF FOUR ROOM HOUSE.
40  DRAW "BM0,0; D50; L100; U50; R50; D50;
    L50; U25; R100
45  REM  ALLOWS FOR DETECTION OF MORE THAN ONE
    ROOM ALARM IN 10 SECONDS.
50  FOR I = 1 TO 3750
60  NEXT I
65  REM  MONITORING THE ALARM DECODER.
70  LET P = JOYSTK(0)
80  LET T = P
85  REM  ALARM ANNUNCIATES.
90  IF T > 0 THEN MOTOR ON ELSE 50
95  REM  COLORS THE ROOMS THAT HAVE AN INTRUSION.
100  IF 33 < = T THEN T = T - 33 ELSE 120
110  PAINT (25,12),4,8
120  IF 16 < = T THEN T = T - 16 ELSE 140
130  PAINT (25,30),4,8
140  IF 7 < = T THEN T = T - 7 ELSE 160
150  PAINT (75,12),4,8
160  IF 3 < = T THEN PAINT (75,30),4,8
165  REM  WAITING TO SILENCE ALARM.
170  LET A$ = INKEY$
180  IF A$ = "S" THEN MOTOR OFF ELSE 170
190  REM  AFTER ALL ROOMS HAVE BEEN CHECKED, TYPE "C"
     TO RESET.
200  LET A$ = INKEY$
210  IF A$ = "C" THEN 10 ELSE 200
```



SYSTEM DRAWING
FIG. 1



POWER SUPPLY/ALARM CODER
FIG. 2

INTRUSION ALARM

PARTS LIST

		RADIO SHACK NO.
RESISTORS		
R1,R2,R3,R4	22K OHM ¼ WATT	271 - 1339
R5,R6,R7,R8, R9,R10,R11,R12,R13,R14,R15	10K OHM ¼ WATT	271 - 1335
R16,R17,R18,R19,R20	4.7K OHM ¼ WATT	271 - 1330
CAPACITORS		
C1	1000uF 35V ELECTROLYTIC	272 - 1032
SEMICONDUCTORS		
Q1,Q2,Q3,Q4	TRANSISTORS 2N3906	276 - 1604
ICs		
IC1	HEXINVERTER 4069 OR 4049	276 - 2449
IC2	+ 5 VOLT REGULATOR 7805	276 - 1770
IC3	1-AMP 50-PIV DIODE BRIDGE	276 - 1161
MISCELLANEOUS		
5	DOOR/WINDOW SWITCHES (CONTACTS OPEN AS MAGNET ENGAGES)	49 - 512
J3,J2	6 PIN DIN-CONNECTORS	274 - 020
T1	12 VAC TRANSFORMER	273 - 1385



HEX INVERTER



HEX INVERTER BUFFER

VOLTAGE REGULATOR 7805

TO-220





TRANSISTOR BOTTOM VIEW



TL082 OP-AMP

PIN OUT